Rules for integrands of the form $u\,(a+b\,\text{ArcSech}[c+d\,x])^p$

1. $\int (a+b\,\text{ArcSech}[c+d\,x])^p\,dx$

   1. $\int \text{ArcSech}[c+d\,x]\,dx$

      1: $\int \text{ArcSech}[c+d\,x]\,dx$

- Reference: CRC 591, A&S 4.6.47
- Derivation: Integration by parts
- Basis: $\partial_x \text{ArcSech}[c+d\,x] == -\frac{d\sqrt{\frac{1-c-d\,x}{1+c+d\,x}}}{(c+d\,x)\,(1-c-d\,x)}$
- Rule:

$$\int \text{ArcSech}[c+d\,x]\,dx \longrightarrow \frac{(c+d\,x)\,\text{ArcSech}[c+d\,x]}{d}+\int \frac{\sqrt{\frac{1-c-d\,x}{1+c+d\,x}}}{1-c-d\,x}\,dx$$

- Program code:

```
Int[ArcSech[c_+d_.*x_],x_Symbol] :=
  (c+d*x)*ArcSech[c+d*x]/d +
  Int[Sqrt[(1-c-d*x)/(1+c+d*x)]/(1-c-d*x),x] /;
FreeQ[{c,d},x]
```

2: $\int \text{ArcCsch}[c + d\,x]\,dx$

- **Reference: CRC 594, A&S 4.6.46**
- **Derivation: Integration by parts**
- **Basis:** $\partial_x \text{ArcCsch}[c + d\,x] == -\frac{d}{(c+d\,x)^2\sqrt{1+\frac{1}{(c+d\,x)^2}}}$
- **Rule:**

$$\int \text{ArcCsch}[c + d\,x]\,dx \longrightarrow \frac{(c + d\,x)\,\text{ArcCsch}[c + d\,x]}{d} + \int \frac{1}{(c + d\,x)\,\sqrt{1 + \frac{1}{(c+d\,x)^2}}}\,dx$$

- **Program code:**

```
Int[ArcCsch[c_+d_.*x_],x_Symbol] :=
  (c+d*x)*ArcCsch[c+d*x]/d +
  Int[1/((c+d*x)*Sqrt[1+1/(c+d*x)^2]),x] /;
FreeQ[{c,d},x]
```

2: $\int (a + b\,\text{ArcSech}[c + d\,x])^p\,dx$ when $p \in \mathbb{Z}^+$

Derivation: Integration by substitution

Rule: If $p \in \mathbb{Z}^+$**, then**

$$\int (a + b\,\text{ArcSech}[c + d\,x])^p\,dx \longrightarrow \frac{1}{d}\,\text{Subst}\Big[\int (a + b\,\text{ArcSech}[x])^p\,dx,\ x,\ c + d\,x\Big]$$

Program code:

```
Int[(a_.+b_.*ArcSech[c_+d_.*x_])^p_.,x_Symbol] :=
  1/d*Subst[Int[(a+b*ArcSech[x])^p,x],x,c+d*x] /;
FreeQ[{a,b,c,d},x] && IGtQ[p,0]
```

```
Int[(a_.+b_.*ArcCsch[c_+d_.*x_])^p_.,x_Symbol] :=
  1/d*Subst[Int[(a+b*ArcCsch[x])^p,x],x,c+d*x] /;
FreeQ[{a,b,c,d},x] && IGtQ[p,0]
```

U: $\int (a + b\,\text{ArcSech}[c + d\,x])^p\,dx$ when $p \notin \mathbb{Z}^+$

Rule: If $p \notin \mathbb{Z}^+$, then

$$\int (a + b\,\text{ArcSech}[c + d\,x])^p\,dx \longrightarrow \int (a + b\,\text{ArcSech}[c + d\,x])^p\,dx$$

Program code:

```
Int[(a_.+b_.*ArcSech[c_+d_.*x_])^p_,x_Symbol] :=
  Unintegrable[(a+b*ArcSech[c+d*x])^p,x] /;
FreeQ[{a,b,c,d,p},x] && Not[IGtQ[p,0]]
```

```
Int[(a_.+b_.*ArcCsch[c_+d_.*x_])^p_,x_Symbol] :=
  Unintegrable[(a+b*ArcCsch[c+d*x])^p,x] /;
FreeQ[{a,b,c,d,p},x] && Not[IGtQ[p,0]]
```

2. $\int (e + f\,x)^m\,(a + b\,\text{ArcSech}[c + d\,x])^p\,dx$

1: $\int (e + f\,x)^m\,(a + b\,\text{ArcSech}[c + d\,x])^p\,dx$ when $d\,e - c\,f == 0 \land p \in \mathbb{Z}^+$

Derivation: Integration by substitution

Rule: If $d\,e - c\,f == 0 \land p \in \mathbb{Z}^+$, then

$$\int (e + f\,x)^m\,(a + b\,\text{ArcSech}[c + d\,x])^p\,dx \longrightarrow \frac{1}{d}\,\text{Subst}\left[\int \left(\frac{f\,x}{d}\right)^m\,(a + b\,\text{ArcSech}[x])^p\,dx,\ x,\ c + d\,x\right]$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*(a_.+b_.*ArcSech[c_+d_.*x_])^p_.,x_Symbol] :=
  1/d*Subst[Int[(f*x/d)^m*(a+b*ArcSech[x])^p,x],x,c+d*x] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[d*e-c*f,0] && IGtQ[p,0]
```

```
Int[(e_.+f_.*x_)^m_.*(a_.+b_.*ArcCsch[c_+d_.*x_])^p_.,x_Symbol] :=
  1/d*Subst[Int[(f*x/d)^m*(a+b*ArcCsch[x])^p,x],x,c+d*x] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[d*e-c*f,0] && IGtQ[p,0]
```

x. $\int x^m \text{ArcSech}[a+b\,x]\,dx$ when $m\in\mathbb{Z}$?? ??

1: $\int x^m \text{ArcSech}[a+b\,x]\,dx$ when $m\in\mathbb{Z}\,\wedge\,m\neq-1$

Derivation: Integration by parts and substitution

- **Basis:** $x^m == -\partial_x \frac{(-a)^{m+1}-b^{m+1}x^{m+1}}{b^{m+1}(m+1)}$
- **Basis: If** $m\in\mathbb{Z}$, **then** $\left((-a)^{m+1}-b^{m+1}x^{m+1}\right)F\left[\frac{1}{a+b\,x}\right] == -\frac{1}{b}\,\text{Subst}\left[\frac{(-a\,x)^{m+1}-(1-a\,x)^{m+1}}{x^{m+3}}F[x],\,x,\,\frac{1}{a+b\,x}\right]\partial_x\frac{1}{a+b\,x}$

Rule: If $m\in\mathbb{Z}\,\wedge\,m\neq-1$, **then**

$$\int x^m \text{ArcSech}[a+b\,x]\,dx \longrightarrow -\frac{\left((-a)^{m+1}-b^{m+1}x^{m+1}\right)\text{ArcSech}[a+b\,x]}{b^{m+1}(m+1)}-\frac{1}{b^m(m+1)}\int\frac{\left((-a)^{m+1}-b^{m+1}x^{m+1}\right)\sqrt{\frac{1-a-b\,x}{1+a+b\,x}}}{(1-a-b\,x)(a+b\,x)}\,dx$$

$$\longrightarrow -\frac{\left((-a)^{m+1}-b^{m+1}x^{m+1}\right)\text{ArcSech}[a+b\,x]}{b^{m+1}(m+1)}+\frac{1}{b^{m+1}(m+1)}\,\text{Subst}\left[\int\frac{\left((-a\,x)^{m+1}-(1-a\,x)^{m+1}\right)}{x^{m+1}\sqrt{-1+x}\,\sqrt{1+x}}\,dx,\,x,\,\frac{1}{a+b\,x}\right]$$

Program code:

```
(* Int[x_^m_.*ArcSech[a_+b_.*x_],x_Symbol] :=
  -((-a)^(m+1)-b^(m+1)*x^(m+1))*ArcSech[a+b*x]/(b^(m+1)*(m+1)) +
  1/(b^(m+1)*(m+1))*Subst[Int[((-a*x)^(m+1)-(1-a*x)^(m+1))/(x^(m+1)*Sqrt[-1+x]*Sqrt[1+x]),x],x,1/(a+b*x)] /;
FreeQ[{a,b},x] && IntegerQ[m] && NeQ[m,-1] *)
```

2: $\int x^m \text{ArcCsch}[a+b\,x]\,dx$ when $m\in\mathbb{Z}\,\wedge\,m\neq-1$

Derivation: Integration by parts and substitution

- **Basis: If** $m\in\mathbb{Z}$, **then** $\frac{\left((-a)^{m+1}-b^{m+1}x^{m+1}\right)}{(a+b\,x)^2}F\left[\frac{1}{a+b\,x}\right] == -\frac{1}{b}\,\text{Subst}\left[\frac{(-a\,x)^{m+1}-(1-a\,x)^{m+1}}{x^{m+1}}F[x],\,x,\,\frac{1}{a+b\,x}\right]\partial_x\frac{1}{a+b\,x}$

Rule: If $m\in\mathbb{Z}\,\wedge\,m\neq-1$, **then**

$$\int x^m \text{ArcCsch}[a+b\,x]\,dx \longrightarrow -\frac{\left((-a)^{m+1}-b^{m+1}x^{m+1}\right)\text{ArcCsch}[a+b\,x]}{b^{m+1}(m+1)}-\frac{1}{b^m(m+1)}\int\frac{\left((-a)^{m+1}-b^{m+1}x^{m+1}\right)}{(a+b\,x)^2\sqrt{1+\frac{1}{(a+b\,x)^2}}}\,dx$$

$$\longrightarrow -\frac{\left((-a)^{m+1}-b^{m+1}x^{m+1}\right)\text{ArcCsch}[a+b\,x]}{b^{m+1}\,(m+1)}+\frac{1}{b^{m+1}\,(m+1)}\,\text{Subst}\left[\int\frac{(-a\,x)^{m+1}-(1-a\,x)^{m+1}}{x^{m+1}\sqrt{1+x^2}}\,dx,\,x,\,\frac{1}{a+b\,x}\right]$$

Program code:

```
(* Int[x_^m_.*ArcCsch[a_+b_.*x_],x_Symbol] :=
  -((-a)^(m+1)-b^(m+1)*x^(m+1))*ArcCsch[a+b*x]/(b^(m+1)*(m+1)) +
  1/(b^(m+1)*(m+1))*Subst[Int[((-a*x)^(m+1)-(1-a*x)^(m+1))/(x^(m+1)*Sqrt[1+x^2]),x],x,1/(a+b*x)] /;
FreeQ[{a,b},x] && IntegerQ[m] && NeQ[m,-1] *)
```

2: $\int (e+f\,x)^m\,(a+b\,\text{ArcSech}[c+d\,x])^p\,dx$ when $p\in\mathbb{Z}^+\wedge m\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: If $m\in\mathbb{Z}$, **then**

$$(e+f\,x)^m\,F[\text{ArcSech}[c+d\,x]] = -\frac{1}{d^{m+1}}\,\text{Subst}[F[x]\,\text{Sech}[x]\,\text{Tanh}[x]\,(d\,e-c\,f+f\,\text{Sech}[x])^m,\,x,\,\text{ArcSech}[c+d\,x]]\,\partial_x\text{ArcSech}[c+d\,x]$$

Rule: If $p\in\mathbb{Z}^+\wedge m\in\mathbb{Z}$, **then**

$$\int (e+f\,x)^m\,(a+b\,\text{ArcSech}[c+d\,x])^p\,dx \longrightarrow -\frac{1}{d^{m+1}}\,\text{Subst}\left[\int (a+b\,x)^p\,\text{Sech}[x]\,\text{Tanh}[x]\,(d\,e-c\,f+f\,\text{Sech}[x])^m\,dx,\,x,\,\text{ArcSech}[c+d\,x]\right]$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*(a_.+b_.*ArcSech[c_+d_.*x_])^p_.,x_Symbol] :=
  -1/d^(m+1)*Subst[Int[(a+b*x)^p*Sech[x]*Tanh[x]*(d*e-c*f+f*Sech[x])^m,x],x,ArcSech[c+d*x]] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[p,0] && IntegerQ[m]
```

```
Int[(e_.+f_.*x_)^m_.*(a_.+b_.*ArcCsch[c_+d_.*x_])^p_.,x_Symbol] :=
  -1/d^(m+1)*Subst[Int[(a+b*x)^p*Csch[x]*Coth[x]*(d*e-c*f+f*Csch[x])^m,x],x,ArcCsch[c+d*x]] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[p,0] && IntegerQ[m]
```

3: $\int (e + f x)^m (a + b \, \text{ArcSech}[c + d x])^p \, dx$ when $p \in \mathbb{Z}^+$

Derivation: Integration by substitution

Rule: If $p \in \mathbb{Z}^+$, then

$$\int (e + f x)^m (a + b \, \text{ArcSech}[c + d x])^p \, dx \longrightarrow \frac{1}{d} \, \text{Subst}\left[\int \left(\frac{d e - c f}{d} + \frac{f x}{d}\right)^m (a + b \, \text{ArcSech}[x])^p \, dx, \, x, \, c + d x\right]$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*(a_.+b_.*ArcSech[c_+d_.*x_])^p_.,x_Symbol] :=
  1/d*Subst[Int[((d*e-c*f)/d+f*x/d)^m*(a+b*ArcSech[x])^p,x],x,c+d*x] /;
FreeQ[{a,b,c,d,e,f,m},x] && IGtQ[p,0]
```

```
Int[(e_.+f_.*x_)^m_.*(a_.+b_.*ArcCsch[c_+d_.*x_])^p_.,x_Symbol] :=
  1/d*Subst[Int[((d*e-c*f)/d+f*x/d)^m*(a+b*ArcCsch[x])^p,x],x,c+d*x] /;
FreeQ[{a,b,c,d,e,f,m},x] && IGtQ[p,0]
```

U: $\int (e + f x)^m (a + b \, \text{ArcSech}[c + d x])^p \, dx$ when $p \notin \mathbb{Z}^+$

Rule: If $p \notin \mathbb{Z}^+$, then

$$\int (e + f x)^m (a + b \, \text{ArcSech}[c + d x])^p \, dx \longrightarrow \int (e + f x)^m (a + b \, \text{ArcSech}[c + d x])^p \, dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*(a_.+b_.*ArcSech[c_+d_.*x_])^p_,x_Symbol] :=
  Unintegrable[(e+f*x)^m*(a+b*ArcSech[c+d*x])^p,x] /;
FreeQ[{a,b,c,d,e,f,m,p},x] && Not[IGtQ[p,0]]
```

```
Int[(e_.+f_.*x_)^m_.*(a_.+b_.*ArcCsch[c_+d_.*x_])^p_,x_Symbol] :=
  Unintegrable[(e+f*x)^m*(a+b*ArcCsch[c+d*x])^p,x] /;
FreeQ[{a,b,c,d,e,f,m,p},x] && Not[IGtQ[p,0]]
```

Rules for integrands involving inverse hyperbolic secants and cosecants

1: $\int u\,\text{ArcSech}\left[\frac{c}{a+b\,x^n}\right]^m dx$

Derivation: Algebraic simplification

- **Basis:** $\text{ArcSech}[z] == \text{ArcCosh}\left[\frac{1}{z}\right]$

Rule:

$$\int u\,\text{ArcSech}\left[\frac{c}{a+b\,x^n}\right]^m dx \longrightarrow \int u\,\text{ArcCosh}\left[\frac{a}{c}+\frac{b\,x^n}{c}\right]^m dx$$

Program code:

```
Int[u_.*ArcSech[c_./(a_.+b_.*x_^n_.)]^m_.,x_Symbol] :=
  Int[u*ArcCosh[a/c+b*x^n/c]^m,x] /;
FreeQ[{a,b,c,n,m},x]
```

```
Int[u_.*ArcCsch[c_./(a_.+b_.*x_^n_.)]^m_.,x_Symbol] :=
  Int[u*ArcSinh[a/c+b*x^n/c]^m,x] /;
FreeQ[{a,b,c,n,m},x]
```

2. $\int v\, e^{\text{ArcSech}[u]}\, dx$

1. $\int e^{n\,\text{ArcSech}[u]}\, dx$

1. $\int e^{\text{ArcSech}[a\,x^p]}\, dx$

1. $\int e^{\text{ArcSech}[a\,x^p]}\, dx$

1: $\int e^{\text{ArcSech}[a\,x]}\, dx$

Derivation: Integration by parts

- **Basis:** $\partial_x e^{\text{ArcSech}[a\,x]} = -\frac{1}{a\,x^2} - \frac{1}{a\,x^2\,(1-a\,x)}\sqrt{\frac{1-a\,x}{1+a\,x}}$

- **Rule:**

$$\int e^{\text{ArcSech}[a\,x]}\, dx \longrightarrow x\, e^{\text{ArcSech}[a\,x]} + \frac{\text{Log}[x]}{a} + \frac{1}{a}\int \frac{1}{x\,(1-a\,x)}\sqrt{\frac{1-a\,x}{1+a\,x}}\, dx$$

- **Program code:**

```
Int[E^ArcSech[a_.*x_], x_Symbol] :=
  x*E^ArcSech[a*x] + Log[x]/a + 1/a*Int[1/(x*(1-a*x))*Sqrt[(1-a*x)/(1+a*x)],x] /;
FreeQ[a,x]
```

2: $\int e^{\text{ArcSech}[a\,x^p]}\,dx$

Derivation: Integration by parts, piecewise constant extraction and algebraic simplification

- **Basis:** $\partial_x e^{\text{ArcSech}[a\,x^p]} == -\frac{p}{a\,x^{p+1}} - \frac{p}{a\,x^{p+1}\,(1-a\,x^p)}\sqrt{\frac{1-a\,x^p}{1+a\,x^p}}$
- **Basis:** $\partial_x\left(\sqrt{\frac{1-a\,x^p}{1+a\,x^p}} \Big/ \frac{\sqrt{1-a\,x^p}}{\sqrt{1+a\,x^p}}\right) == 0$
- **Basis:** $\sqrt{\frac{1-a\,x^p}{1+a\,x^p}} \Big/ \frac{\sqrt{1-a\,x^p}}{\sqrt{1+a\,x^p}} == \sqrt{1+a\,x^p}\,\sqrt{\frac{1}{1+a\,x^p}}$

Rule:

$$\int e^{\text{ArcSech}[a\,x^p]}\,dx \longrightarrow x\,e^{\text{ArcSech}[a\,x^p]} + \frac{p}{a}\int\frac{1}{x^p}\,dx + \frac{p\sqrt{1+a\,x^p}}{a}\sqrt{\frac{1}{1+a\,x^p}}\int\frac{1}{x^p\sqrt{1+a\,x^p}\,\sqrt{1-a\,x^p}}\,dx$$

Program code:

```
Int[E^ArcSech[a_.*x_^p_], x_Symbol] :=
  x*E^ArcSech[a*x^p] +
  p/a*Int[1/x^p,x] +
  p*Sqrt[1+a*x^p]/a*Sqrt[1/(1+a*x^p)]*Int[1/(x^p*Sqrt[1+a*x^p]*Sqrt[1-a*x^p]),x] /;
FreeQ[{a,p},x]
```

2: $\int e^{\text{ArcCsch}[a\,x^p]}\,dx$

Derivation: Algebraic simplification

- **Basis:** $e^{\text{ArcCsch}[z]} == \frac{1}{z} + \sqrt{1 + \frac{1}{z^2}}$

Rule:

$$\int e^{\text{ArcCsch}[a\,x^p]}\,dx \longrightarrow \frac{1}{a}\int \frac{1}{x^p}\,dx + \int \sqrt{1 + \frac{1}{a^2\,x^{2\,p}}}\,dx$$

Program code:

```
Int[E^ArcCsch[a_.*x_^p_.], x_Symbol] :=
  1/a*Int[1/x^p,x] + Int[Sqrt[1+1/(a^2*x^(2*p))],x] /;
FreeQ[{a,p},x]
```

2. $\int e^{n \text{ArcSech}[u]} \, dx$ when $n \in \mathbb{Z}$

1: $\int e^{n \text{ArcSech}[u]} \, dx$ when $n \in \mathbb{Z}$

Derivation: Algebraic simplification

- **Basis:** $e^{\text{ArcSech}[z]} == \frac{1}{z} + \frac{1+z}{z} \sqrt{\frac{1-z}{1+z}} == \frac{1}{z} + \sqrt{\frac{1-z}{1+z}} + \frac{1}{z} \sqrt{\frac{1-z}{1+z}}$

- **Basis:** $e^{n \text{ArcSech}[z]} == \left(\frac{1}{z} + \sqrt{-1 + \frac{1}{z}} \sqrt{1 + \frac{1}{z}}\right)^n$

Basis: If $n \in \mathbb{Z}$**, then** $e^{n z} == (e^z)^n$

Rule: If $n \in \mathbb{Z}$**, then**

$$\int e^{n \text{ArcSech}[u]} \, dx \longrightarrow \int \left(\frac{1}{u} + \sqrt{\frac{1-u}{1+u}} + \frac{1}{u} \sqrt{\frac{1-u}{1+u}}\right)^n dx$$

Program code:

```
Int[E^(n_.*ArcSech[u_]), x_Symbol] :=
  Int[(1/u + Sqrt[(1-u)/(1+u)] + 1/u*Sqrt[(1-u)/(1+u)])^n,x] /;
IntegerQ[n]
```

2: $\int e^{n\,\text{ArcCsch}[u]}\,dx$ when $n \in \mathbb{Z}$

Derivation: Algebraic simplification

- **Basis:** $e^{n\,\text{ArcCsch}[z]} = \left(\frac{1}{z} + \sqrt{1 + \frac{1}{z^2}}\right)^n$

Rule: If $n \in \mathbb{Z}$**, then**

$$\int e^{n\,\text{ArcCsch}[u]}\,dx \longrightarrow \int \left(\frac{1}{u} + \sqrt{1 + \frac{1}{u^2}}\right)^n dx$$

Program code:

```
Int[E^(n_.*ArcCsch[u_]), x_Symbol] :=
  Int[(1/u + Sqrt[1+1/u^2])^n,x] /;
IntegerQ[n]
```

2. $\int x^m e^{n\,\text{ArcSech}[u]}\,dx$

   1. $\int x^m e^{\text{ArcSech}[a\,x^p]}\,dx$

      1. $\int x^m e^{\text{ArcSech}[a\,x^p]}\,dx$

         1: $\int \frac{e^{\text{ArcSech}[a\,x^p]}}{x}\,dx$

Derivation: Algebraic simplification, piecewise constant extraction and algebraic simplification

- **Basis:** $e^{\text{ArcSech}[z]} == \frac{1}{z} + \frac{1+z}{z}\sqrt{\frac{1-z}{1+z}} == \frac{1}{z} + \sqrt{\frac{1-z}{1+z}} + \frac{1}{z}\sqrt{\frac{1-z}{1+z}}$

- **Basis:** $\partial_x\left(\sqrt{\frac{1-a\,x^p}{1+a\,x^p}} \Big/ \frac{\sqrt{1-a\,x^p}}{\sqrt{1+a\,x^p}}\right) == 0$

- **Basis:** $\sqrt{\frac{1-a\,x^p}{1+a\,x^p}} \Big/ \frac{\sqrt{1-a\,x^p}}{\sqrt{1+a\,x^p}} == \sqrt{1+a\,x^p}\,\sqrt{\frac{1}{1+a\,x^p}}$

Rule:

$$\int \frac{e^{\text{ArcSech}[a\,x^p]}}{x}\,dx \longrightarrow -\frac{1}{a\,p\,x^p} + \frac{\sqrt{1+a\,x^p}}{a}\sqrt{\frac{1}{1+a\,x^p}}\int \frac{\sqrt{1+a\,x^p}\,\sqrt{1-a\,x^p}}{x^{p+1}}\,dx$$

Program code:

```
Int[E^ArcSech[a_.*x_^p_.]/x_, x_Symbol] :=
  -1/(a*p*x^p) +
  Sqrt[1+a*x^p]/a*Sqrt[1/(1+a*x^p)]*Int[Sqrt[1+a*x^p]*Sqrt[1-a*x^p]/x^(p+1),x] /;
FreeQ[{a,p},x]
```

2: $\int x^m e^{\text{ArcSech}[a x^p]} \, dx$ when $m \neq -1$

Derivation: Integration by parts, piecewise constant extraction and algebraic simplification

- **Basis:** $\partial_x e^{\text{ArcSech}[a x^p]} == -\frac{p}{a x^{p+1}} - \frac{p}{a x^{p+1} (1-a x^p)} \sqrt{\frac{1-a x^p}{1+a x^p}}$

- **Basis:** $\partial_x \left(\sqrt{\frac{1-a x^p}{1+a x^p}} \Big/ \frac{\sqrt{1-a x^p}}{\sqrt{1+a x^p}} \right) == 0$

- **Basis:** $\sqrt{\frac{1-a x^p}{1+a x^p}} \Big/ \frac{\sqrt{1-a x^p}}{\sqrt{1+a x^p}} == \sqrt{1+a x^p} \sqrt{\frac{1}{1+a x^p}}$

Rule: If $m \neq -1$**, then**

$$\int x^m e^{\text{ArcSech}[a x^p]} \, dx \longrightarrow \frac{x^{m+1} e^{\text{ArcSech}[a x^p]}}{m+1} + \frac{p}{a (m+1)} \int x^{m-p} \, dx + \frac{p \sqrt{1+a x^p}}{a (m+1)} \sqrt{\frac{1}{1+a x^p}} \int \frac{x^{m-p}}{\sqrt{1+a x^p} \sqrt{1-a x^p}} \, dx$$

Program code:

```
Int[x_^m_.*E^ArcSech[a_.*x_^p_.], x_Symbol] :=
  x^(m+1)*E^ArcSech[a*x^p]/(m+1) +
  p/(a*(m+1))*Int[x^(m-p),x] +
  p*Sqrt[1+a*x^p]/(a*(m+1))*Sqrt[1/(1+a*x^p)]*Int[x^(m-p)/(Sqrt[1+a*x^p]*Sqrt[1-a*x^p]),x] /;
FreeQ[{a,m,p},x] && NeQ[m,-1]
```

2: $\int x^m e^{\text{ArcCsch}[a x^p]} \, dx$

Derivation: Algebraic simplification

- **Basis:** $e^{\text{ArcCsch}[z]} == \frac{1}{z} + \sqrt{1 + \frac{1}{z^2}}$

Rule:

$$\int x^m e^{\text{ArcCsch}[a x^p]} \, dx \longrightarrow \frac{1}{a} \int x^{m-p} \, dx + \int x^m \sqrt{1 + \frac{1}{a^2 x^{2p}}} \, dx$$

Program code:

```
Int[x_^m_.*E^ArcCsch[a_.*x_^p_.], x_Symbol] :=
  1/a*Int[x^(m-p),x] + Int[x^m*Sqrt[1+1/(a^2*x^(2*p))],x] /;
FreeQ[{a,m,p},x]
```

2. $\int x^m e^{n\,\text{ArcSech}[u]}\,dx$

1: $\int x^m e^{n\,\text{ArcSech}[u]}\,dx$ when $n \in \mathbb{Z}$

Derivation: Algebraic simplification

- **Basis:** $e^{\text{ArcSech}[z]} == \frac{1}{z} + \frac{1+z}{z}\sqrt{\frac{1-z}{1+z}} == \frac{1}{z} + \sqrt{\frac{1-z}{1+z}} + \frac{1}{z}\sqrt{\frac{1-z}{1+z}}$

- **Basis:** $e^{n\,\text{ArcSech}[z]} == \left(\frac{1}{z} + \sqrt{-1+\frac{1}{z}}\,\sqrt{1+\frac{1}{z}}\right)^n$

Rule: If $n \in \mathbb{Z}$**, then**

$$\int x^m e^{n\,\text{ArcSech}[u]}\,dx \longrightarrow \int x^m \left(\frac{1}{u} + \sqrt{\frac{1-u}{1+u}} + \frac{1}{u}\sqrt{\frac{1-u}{1+u}}\right)^n dx$$

Program code:

```
Int[x_^m_.*E^(n_.*ArcSech[u_]), x_Symbol] :=
  Int[x^m*(1/u + Sqrt[(1-u)/(1+u)] + 1/u*Sqrt[(1-u)/(1+u)])^n,x] /;
FreeQ[m,x] && IntegerQ[n]
```

2: $\int x^m e^{n \text{ArcCsch}[u]} \, dx$ when $n \in \mathbb{Z}$

Derivation: Algebraic simplification

- **Basis:** $e^{n \text{ArcCsch}[z]} == \left(\frac{1}{z} + \sqrt{1 + \frac{1}{z^2}}\right)^n$

Rule: If $n \in \mathbb{Z}$, then

$$\int x^m e^{n \text{ArcCsch}[u]} \, dx \longrightarrow \int x^m \left(\frac{1}{u} + \sqrt{1 + \frac{1}{u^2}}\right)^n dx$$

Program code:

```
Int[x_^m_.*E^(n_.*ArcCsch[u_]), x_Symbol] :=
  Int[x^m*(1/u + Sqrt[1+1/u^2])^n,x] /;
FreeQ[m,x] && IntegerQ[n]
```

3: $\int \frac{e^{\text{ArcSech}[c\,x]}}{a+b\,x^2}\,dx$ when $b+a\,c^2==0$

Derivation: Algebraic expansion

- **Basis:** $\frac{e^{\text{ArcSech}[x]}}{1-x^2}==\frac{\sqrt{\frac{1}{1+x}}}{x\,\sqrt{1-x}}+\frac{1}{x\,(1-x^2)}$

- **Basis: If** $b+a\,c^2==0$ **, then** $\frac{e^{\text{ArcSech}[c\,x]}}{a+b\,x^2}==\frac{\sqrt{\frac{1}{1+c\,x}}}{a\,c\,x\,\sqrt{1-c\,x}}+\frac{1}{c\,x\,(a+b\,x^2)}$

- **Rule: If** $b+a\,c^2==0$ **, then**

$$\int \frac{e^{\text{ArcSech}[c\,x]}}{a+b\,x^2}\,dx \longrightarrow \frac{1}{a\,c}\int \frac{\sqrt{\frac{1}{1+c\,x}}}{x\,\sqrt{1-c\,x}}\,dx+\frac{1}{c}\int \frac{1}{x\,\left(a+b\,x^2\right)}\,dx$$

- **Basis:** $\frac{e^{\text{ArcCsch}[x]}}{1+x^2}==\frac{1}{x^2\,\sqrt{1+\frac{1}{x^2}}}+\frac{1}{x\,(1+x^2)}$

- **Basis: If** $b-a\,c^2==0$ **, then** $\frac{e^{\text{ArcCsch}[c\,x]}}{a+b\,x^2}==\frac{1}{a\,c^2\,x^2\,\sqrt{1+\frac{1}{c^2\,x^2}}}+\frac{1}{c\,x\,(a+b\,x^2)}$

- **Rule: If** $b-a\,c^2==0$ **, then**

$$\int \frac{e^{\text{ArcCsch}[c\,x]}}{a+b\,x^2}\,dx \longrightarrow \frac{1}{a\,c^2}\int \frac{1}{x^2\,\sqrt{1+\frac{1}{c^2\,x^2}}}\,dx+\frac{1}{c}\int \frac{1}{x\,\left(a+b\,x^2\right)}\,dx$$

Program code:

```
Int[E^(ArcSech[c_.*x_])/(a_+b_.*x_^2), x_Symbol] :=
  1/(a*c)*Int[Sqrt[1/(1+c*x)]/(x*Sqrt[1-c*x]),x] + 1/c*Int[1/(x*(a+b*x^2)),x] /;
FreeQ[{a,b,c},x] && EqQ[b+a*c^2,0]
```

```
Int[E^(ArcCsch[c_.*x_])/(a_+b_.*x_^2), x_Symbol] :=
  1/(a*c^2)*Int[1/(x^2*Sqrt[1+1/(c^2*x^2)]),x] + 1/c*Int[1/(x*(a+b*x^2)),x] /;
FreeQ[{a,b,c},x] && EqQ[b-a*c^2,0]
```

4: $\int \frac{(d\,x)^m\,e^{\text{ArcSech}[c\,x]}}{a+b\,x^2}\,dx$ when $b+a\,c^2=0$

- **Derivation: Algebraic expansion**

- **Basis:** $\frac{e^{\text{ArcSech}[x]}}{1-x^2}=\frac{\sqrt{\frac{1}{1+x}}}{x\,\sqrt{1-x}}+\frac{1}{x\,(1-x^2)}$

- **Basis: If** $b+a\,c^2=0$, **then** $\frac{(d\,x)^m\,e^{\text{ArcSech}[c\,x]}}{a+b\,x^2}=\frac{d\,(d\,x)^{m-1}\sqrt{\frac{1}{1+c\,x}}}{a\,c\,\sqrt{1-c\,x}}+\frac{d\,(d\,x)^{m-1}}{c\,(a+b\,x^2)}$

- **Rule: If** $b+a\,c^2=0$, **then**

$$\int \frac{(d\,x)^m\,e^{\text{ArcSech}[c\,x]}}{a+b\,x^2}\,dx\,\rightarrow\,\frac{d}{a\,c}\int \frac{(d\,x)^{m-1}\sqrt{\frac{1}{1+c\,x}}}{\sqrt{1-c\,x}}\,dx+\frac{d}{c}\int \frac{(d\,x)^{m-1}}{a+b\,x^2}\,dx$$

- **Basis:** $\frac{e^{\text{ArcCsch}[x]}}{1+x^2}=\frac{1}{x^2\,\sqrt{1+\frac{1}{x^2}}}+\frac{1}{x\,(1+x^2)}$

- **Basis: If** $b-a\,c^2=0$, **then** $\frac{(d\,x)^m\,e^{\text{ArcCsch}[c\,x]}}{a+b\,x^2}=\frac{d^2\,(d\,x)^{m-2}}{a\,c^2\,\sqrt{1+\frac{1}{c^2\,x^2}}}+\frac{d\,(d\,x)^{m-1}}{c\,(a+b\,x^2)}$

- **Rule: If** $b-a\,c^2=0$, **then**

$$\int \frac{(d\,x)^m\,e^{\text{ArcCsch}[c\,x]}}{a+b\,x^2}\,dx\,\rightarrow\,\frac{d^2}{a\,c^2}\int \frac{(d\,x)^{m-2}}{\sqrt{1+\frac{1}{c^2\,x^2}}}\,dx+\frac{d}{c}\int \frac{(d\,x)^{m-1}}{a+b\,x^2}\,dx$$

Program code:

```
Int[(d_.*x_)^m_.*E^(ArcSech[c_.*x_])/(a_+b_.*x_^2), x_Symbol] :=
  d/(a*c)*Int[(d*x)^(m-1)*Sqrt[1/(1+c*x)]/Sqrt[1-c*x],x] + d/c*Int[(d*x)^(m-1)/(a+b*x^2),x] /;
FreeQ[{a,b,c,d,m},x] && EqQ[b+a*c^2,0]
```

```
Int[(d_.*x_)^m_.*E^(ArcCsch[c_.*x_])/(a_+b_.*x_^2), x_Symbol] :=
  d^2/(a*c^2)*Int[(d*x)^(m-2)/Sqrt[1+1/(c^2*x^2)],x] + d/c*Int[(d*x)^(m-1)/(a+b*x^2),x] /;
FreeQ[{a,b,c,d,m},x] && EqQ[b-a*c^2,0]
```

3. $\int v\,(a+b\,\text{ArcSech}[u])\,dx$ when **u is free of inverse functions**

1. $\int \text{ArcSech}[u]\,dx$ when **u is free of inverse functions**

1: $\int \text{ArcSech}[u]\,dx$ when **u is free of inverse functions**

Derivation: Integration by parts and piecewise constant extraction

- **Basis:** $\partial_x \text{ArcSech}[f[x]] == -\frac{\partial_x f[x]}{f[x]^2\sqrt{-1+\frac{1}{f[x]}}\sqrt{1+\frac{1}{f[x]}}}$

- **Basis:** $\partial_x \frac{\sqrt{1-f[x]^2}}{f[x]\sqrt{-1+\frac{1}{f[x]}}\sqrt{1+\frac{1}{f[x]}}} == 0$

Rule: If u is free of inverse functions, then

$$\int \text{ArcSech}[u]\,dx \longrightarrow x\,\text{ArcSech}[u]+\int \frac{x\,\partial_x u}{u^2\sqrt{-1+\frac{1}{u}}\sqrt{1+\frac{1}{u}}}\,dx \longrightarrow x\,\text{ArcSech}[u]+\frac{\sqrt{1-u^2}}{u\sqrt{-1+\frac{1}{u}}\sqrt{1+\frac{1}{u}}}\int \frac{x\,\partial_x u}{u\sqrt{1-u^2}}\,dx$$

Program code:

```
Int[ArcSech[u_],x_Symbol] :=
  x*ArcSech[u] +
  Sqrt[1-u^2]/(u*Sqrt[-1+1/u]*Sqrt[1+1/u])*Int[SimplifyIntegrand[x*D[u,x]/(u*Sqrt[1-u^2]),x],x] /;
InverseFunctionFreeQ[u,x] && Not[FunctionOfExponentialQ[u,x]]
```

2: $\int \text{ArcCsch}[u]\,dx$ when **u is free of inverse functions**

Derivation: Integration by parts and piecewise constant extraction

- **Basis:** $\partial_x \text{ArcCsch}[f[x]] == -\frac{\partial_x f[x]}{f[x]^2\sqrt{1+\frac{1}{f[x]^2}}} == \frac{\partial_x f[x]}{\sqrt{-f[x]^2}\sqrt{-1-f[x]^2}}$

- **Basis:** $\partial_x \frac{f[x]}{\sqrt{-f[x]^2}} == 0$

Rule: If u is free of inverse functions, then

$$\int \text{ArcCsch}[u]\, dx \longrightarrow x\,\text{ArcCsch}[u] - \int \frac{x\,\partial_x u}{\sqrt{-u^2}\,\sqrt{-1-u^2}}\, dx \longrightarrow x\,\text{ArcCsch}[u] - \frac{u}{\sqrt{-u^2}} \int \frac{x\,\partial_x u}{u\,\sqrt{-1-u^2}}\, dx$$

Program code:

```
Int[ArcCsch[u_],x_Symbol] :=
  x*ArcCsch[u] -
  u/Sqrt[-u^2]*Int[SimplifyIntegrand[x*D[u,x]/(u*Sqrt[-1-u^2]),x],x] /;
InverseFunctionFreeQ[u,x] && Not[FunctionOfExponentialQ[u,x]]
```

2. $\int (c+d\,x)^m\,(a+b\,\text{ArcSech}[u])\, dx$ when $m \neq -1 \land u$ is free of inverse functions

1: $\int (c+d\,x)^m\,(a+b\,\text{ArcSech}[u])\, dx$ when $m \neq -1 \land u$ is free of inverse functions

Derivation: Integration by parts and piecewise constant extraction

Basis: $\partial_x \text{ArcSech}[f[x]] == -\frac{\partial_x f[x]}{f[x]^2\,\sqrt{-1+\frac{1}{f[x]}}\,\sqrt{1+\frac{1}{f[x]}}}$

Basis: $\partial_x \frac{\sqrt{1-f[x]^2}}{f[x]\,\sqrt{-1+\frac{1}{f[x]}}\,\sqrt{1+\frac{1}{f[x]}}} == 0$

Rule: If $m \neq -1 \land$ **if u is free of inverse functions, then**

$$\int (c+d\,x)^m\,(a+b\,\text{ArcSech}[u])\, dx \longrightarrow \frac{(c+d\,x)^{m+1}\,(a+b\,\text{ArcSech}[u])}{d\,(m+1)} + \frac{b}{d\,(m+1)} \int \frac{(c+d\,x)^{m+1}\,\partial_x u}{u^2\,\sqrt{-1+\frac{1}{u}}\,\sqrt{1+\frac{1}{u}}}\, dx$$

$$\longrightarrow \frac{(c+d\,x)^{m+1}\,(a+b\,\text{ArcSech}[u])}{d\,(m+1)} + \frac{b\,\sqrt{1-u^2}}{d\,(m+1)\,u\,\sqrt{-1+\frac{1}{u}}\,\sqrt{1+\frac{1}{u}}} \int \frac{(c+d\,x)^{m+1}\,\partial_x u}{u\,\sqrt{1-u^2}}\, dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*(a_.+b_.*ArcSech[u_]),x_Symbol] :=
  (c+d*x)^(m+1)*(a+b*ArcSech[u])/(d*(m+1)) +
  b*Sqrt[1-u^2]/(d*(m+1)*u*Sqrt[-1+1/u]*Sqrt[1+1/u])*Int[SimplifyIntegrand[(c+d*x)^(m+1)*D[u,x]/(u*Sqrt[1-u^2]),x],x] /;
FreeQ[{a,b,c,d,m},x] && NeQ[m,-1] && InverseFunctionFreeQ[u,x] && Not[FunctionOfQ[(c+d*x)^(m+1),u,x]] && Not[FunctionOfExponential(
```

2: $\int (c + d x)^m (a + b \operatorname{ArcCsch}[u]) \, dx$ when $m \neq -1 \land u$ is free of inverse functions

Derivation: Integration by parts and piecewise constant extraction

- **Basis:** $\partial_x (a + b \operatorname{ArcCsch}[f[x]]) = -\frac{b \, \partial_x f[x]}{f[x]^2 \sqrt{1+\frac{1}{f[x]^2}}} = \frac{b \, \partial_x f[x]}{\sqrt{-f[x]^2} \sqrt{-1-f[x]^2}}$

- **Basis:** $\partial_x \frac{f[x]}{\sqrt{-f[x]^2}} = 0$

Rule: If $m \neq -1 \land u$ **is free of inverse functions, then**

$$\int (c + d x)^m (a + b \operatorname{ArcCsch}[u]) \, dx \to \frac{(c + d x)^{m+1} (a + b \operatorname{ArcCsch}[u])}{d (m + 1)} - \frac{b}{d (m + 1)} \int \frac{(c + d x)^{m+1} \partial_x u}{\sqrt{-u^2} \sqrt{-1 - u^2}} \, dx$$

$$\to \frac{(c + d x)^{m+1} (a + b \operatorname{ArcCsch}[u])}{d (m + 1)} - \frac{b u}{d (m + 1) \sqrt{-u^2}} \int \frac{(c + d x)^{m+1} \partial_x u}{u \sqrt{-1 - u^2}} \, dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*(a_.+b_.*ArcCsch[u_]),x_Symbol] :=
  (c+d*x)^(m+1)*(a+b*ArcCsch[u])/(d*(m+1)) -
  b*u/(d*(m+1)*Sqrt[-u^2])*Int[SimplifyIntegrand[(c+d*x)^(m+1)*D[u,x]/(u*Sqrt[-1-u^2]),x],x] /;
FreeQ[{a,b,c,d,m},x] && NeQ[m,-1] && InverseFunctionFreeQ[u,x] && Not[FunctionOfQ[(c+d*x)^(m+1),u,x]] && Not[FunctionOfExponential(
```

3. $\int v (a + b \operatorname{ArcSech}[u]) \, dx$ when u and $\int v \, dx$ are free of inverse functions

1: $\int v (a + b \operatorname{ArcSech}[u]) \, dx$ when u and $\int v \, dx$ are free of inverse functions

Derivation: Integration by parts and piecewise constant extraction

- **Basis:** $\partial_x \operatorname{ArcSech}[f[x]] = -\frac{\partial_x f[x]}{f[x]^2 \sqrt{-1+\frac{1}{f[x]}} \sqrt{1+\frac{1}{f[x]}}}$

- **Basis:** $\partial_x \frac{\sqrt{1-f[x]^2}}{f[x] \sqrt{-1+\frac{1}{f[x]}} \sqrt{1+\frac{1}{f[x]}}} = 0$

- **Rule: If u is free of inverse functions, let** $w = \int v \, dx$**, if w is free of inverse functions, then**

$$\int v\,(a+b\,\text{ArcSech}[u])\,dx \longrightarrow w\,(a+b\,\text{ArcSech}[u])+b\int\frac{w\,\partial_x u}{u^2\sqrt{-1+\frac{1}{u}}\sqrt{1+\frac{1}{u}}}\,dx \longrightarrow w\,(a+b\,\text{ArcSech}[u])+\frac{b\sqrt{1-u^2}}{u\sqrt{-1+\frac{1}{u}}\sqrt{1+\frac{1}{u}}}\int\frac{w\,\partial_x u}{u\sqrt{1-u^2}}\,dx$$

- **Program code:**

```
Int[v_*(a_.+b_.*ArcSech[u_]),x_Symbol] :=
  With[{w=IntHide[v,x]},
  Dist[(a+b*ArcSech[u]),w,x] + b*Sqrt[1-u^2]/(u*Sqrt[-1+1/u]*Sqrt[1+1/u])*Int[SimplifyIntegrand[w*D[u,x]/(u*Sqrt[1-u^2]),x],x] /;
 InverseFunctionFreeQ[w,x]] /;
FreeQ[{a,b},x] && InverseFunctionFreeQ[u,x] && Not[MatchQ[v, (c_.+d_.*x)^m_. /; FreeQ[{c,d,m},x]]]
```

2: $\int v\,(a+b\,\text{ArcCsch}[u])\,dx$ when u and $\int v\,dx$ are free of inverse functions

- **Derivation: Integration by parts and piecewise constant extraction**
- **Basis:** $\partial_x\,(a+b\,\text{ArcCsch}[f[x]]) == -\frac{b\,\partial_x f[x]}{f[x]^2\sqrt{1+\frac{1}{f[x]^2}}} == \frac{b\,\partial_x f[x]}{\sqrt{-f[x]^2}\sqrt{-1-f[x]^2}}$
- **Basis:** $\partial_x\frac{f[x]}{\sqrt{-f[x]^2}} == 0$
- **Rule: If u is free of inverse functions, let** $w == \int v\,dx$**, if w is free of inverse functions, then**

$$\int v\,(a+b\,\text{ArcCsch}[u])\,dx \longrightarrow w\,(a+b\,\text{ArcCsch}[u])-b\int\frac{w\,\partial_x u}{\sqrt{-u^2}\sqrt{-1-u^2}}\,dx \longrightarrow w\,(a+b\,\text{ArcCsch}[u])-\frac{b\,u}{\sqrt{-u^2}}\int\frac{w\,\partial_x u}{u\sqrt{-1-u^2}}\,dx$$

- **Program code:**

```
Int[v_*(a_.+b_.*ArcCsch[u_]),x_Symbol] :=
  With[{w=IntHide[v,x]},
  Dist[(a+b*ArcCsch[u]),w,x] - b*u/Sqrt[-u^2]*Int[SimplifyIntegrand[w*D[u,x]/(u*Sqrt[-1-u^2]),x],x] /;
 InverseFunctionFreeQ[w,x]] /;
FreeQ[{a,b},x] && InverseFunctionFreeQ[u,x] && Not[MatchQ[v, (c_.+d_.*x)^m_. /; FreeQ[{c,d,m},x]]]
```